Appendix B - Derivation of the recursions and key analytic results

Supplement to: Olito et al. (2018) The interaction between sex-specific selection and local adaptation in species without separate sexes. *Phil. Trans. Roy. Soc. Lond. B.*

Colin Olito
[1]Centre for Geometric Biology, School of Biological Sciences, Monash University, Victoria 3800, Australia.
[2]Department of Biology, Section for Evolutionary Ecology, Lund University, Lund, Sweden.

Jessica K. Abbott
[2]Department of Biology, Section for Evolutionary Ecology, Lund University, Lund, Sweden.

Crispin Y. Jordan
[3]University of Edinburgh School of Biomedical Sciences, University of Edinburgh, Edinburgh, Scotland

A 1-locus model of sex-specific selection in the k*th* patch for simultaneous hermaphrodites

```
ClearAll["Global`*"]
```

- QE genotypic frequencies

$$FAA[q\_, C\_] := (1-q)^2 + \frac{C\,q\,(1-q)}{(2-C)};$$

$$FAa[q\_, C\_] := 2\,q\,(1-q) - \frac{2\,C\,q\,(1-q)}{(2-C)};$$

$$Faa[q\_, C\_] := q^2 + \frac{C\,q\,(1-q)}{(2-C)};$$

- Relative contribution of genotypes to offspring via female sex-function

```
FfAA[q_, C_, wfAA_, wfAa_, wfaa_] := (FAA[q, C] * wfAA) /
    (FAA[q, C] * wfAA + FAa[q, C] * wfAa + Faa[q, C] * wfaa);
FfAa[q_, C_, wfAA_, wfAa_, wfaa_] := (FAa[q, C] * wfAa) /
    (FAA[q, C] * wfAA + FAa[q, C] * wfAa + Faa[q, C] * wfaa);
Ffaa[q_, C_, wfAA_, wfAa_, wfaa_] := (Faa[q, C] * wfaa) /
    (FAA[q, C] * wfAA + FAa[q, C] * wfAa + Faa[q, C] * wfaa);
```

Relative contribution of genotypes to offspring via male sex-function

```
FmAA[q_, C_, wmAA_, wmAa_, wmaa_] := (FAA[q, C] * wmAA) /
    (FAA[q, C] * wmAA + FAa[q, C] * wmAa + Faa[q, C] * wmaa);
FmAa[q_, C_, wmAA_, wmAa_, wmaa_] := (FAa[q, C] * wmAa) /
    (FAA[q, C] * wmAA + FAa[q, C] * wmAa + Faa[q, C] * wmaa);
Fmaa[q_, C_, wmAA_, wmAa_, wmaa_] := (Faa[q, C] * wmaa) /
    (FAA[q, C] * wmAA + FAa[q, C] * wmAa + Faa[q, C] * wmaa);
```

Change in allele frequency among female and male gametes

Frequency among ovules

```
qPrf[q_, C_, wfAA_, wfAa_, wfaa_] :=
  Ffaa[q, C, wfAA, wfAa, wfaa] + FfAa[q, C, wfAA, wfAa, wfaa]/2 // FullSimplify
qPrf[q, C, wfAA, wfAa, wfaa] // FullSimplify
```

$(q (-C + 2 (-1 + C) q) wfaa - 2 (-1 + C) (-1 + q) q\, wfAa) / (q (-C + 2 (-1 + C) q) wfaa - 4 (-1 + C) (-1 + q) q\, wfAa + (-1 + q) (2 - C + 2 (-1 + C) q) wfAA)$

Frequency among pollen/sperm

```
qPrm[q_, C_, wmAA_, wmAa_, wmaa_] :=
  Fmaa[q, C, wmAA, wmAa, wmaa] + FmAa[q, C, wmAA, wmAa, wmaa]/2 // FullSimplify
qPrm[q, C, wmAA, wmAa, wmaa] // FullSimplify
```

$(q (-C + 2 (-1 + C) q) wmaa - 2 (-1 + C) (-1 + q) q\, wmAa) / (q (-C + 2 (-1 + C) q) wmaa - 4 (-1 + C) (-1 + q) q\, wmAa + (-1 + q) (2 - C + 2 (-1 + C) q) wmAA)$

Allele frequency in the next generation

```
Wtot[q_, C_, δ_] := 1 - C δ
qPr[q_, C_, δ_, wfAA_, wfAa_, wfaa_, wmAA_, wmAa_, wmaa_] := 1/Wtot[q, C, δ]
  ((1 - C) 1/2 (qPrf[q, C, wfAA, wfAa, wfaa] + qPrm[q, C, wmAA, wmAa, wmaa]) +
    C * (1 - δ) qPrf[q, C, wfAA, wfAa, wfaa])
```

```
qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa]
```

$$\frac{1}{1-C\,\delta}\left(\frac{1}{2}\,(1-C)\,\Big(\big(q\,(-C+2\,(-1+C)\,q)\,\text{wfaa}-2\,(-1+C)\,(-1+q)\,q\,\text{wfAa}\big)\Big/\big(q\,(-C+2\,(-1+C)\,q)\,\text{wfaa}-4\,(-1+C)\,(-1+q)\,q\,\text{wfAa}+(-1+q)\,(2-C+2\,(-1+C)\,q)\,\text{wfAA}\big)+(q\,(-C+2\,(-1+C)\,q)\,\text{wmaa}-2\,(-1+C)\,(-1+q)\,q\,\text{wmAa})\,/\,(q\,(-C+2\,(-1+C)\,q)\,\text{wmaa}-4\,(-1+C)\,(-1+q)\,q\,\text{wmAa}+(-1+q)\,(2-C+2\,(-1+C)\,q)\,\text{wmAA})\Big)+\big(C\,\big(q\,(-C+2\,(-1+C)\,q)\,\text{wfaa}-2\,(-1+C)\,(-1+q)\,q\,\text{wfAa}\big)\,(1-\delta)\big)\Big/\big(q\,(-C+2\,(-1+C)\,q)\,\text{wfaa}-4\,(-1+C)\,(-1+q)\,q\,\text{wfAa}+(-1+q)\,(2-C+2\,(-1+C)\,q)\,\text{wfAA}\big)\right)$$

Stability at the boundary equilibria ($\hat{q}=0$, $\hat{q}=1$)

- Arbitrary fitness expressions (see Table 1 in the main text)

```
λk0 :=
  D[qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa], q] /. q → 0 // FullSimplify
λk1 := D[qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa], q] /. q → 1 //
   FullSimplify
λk0
λk1
```

$$\big(2\,(\text{wfAA wmAa}+\text{wfAa wmAA})+C^2\,\big(-\text{wfAA}\,(\text{wmaa}-2\,\text{wmAa})-(\text{wfaa}-2\,\text{wfAa})\,\text{wmAA}\,(-1+2\,\delta)\big)+C\,\big(\text{wfAA}\,(\text{wmaa}-4\,\text{wmAa})+\text{wmAA}\,(\text{wfaa}-4\,\text{wfAa}\,\delta)\big)\big)\Big/\big(2\,(-2+C)\,\text{wfAA wmAA}\,(-1+C\,\delta)\big)$$

$$\big(2\,\text{wfaa wmAa}+C\,\text{wfaa}\,(2\,(-2+C)\,\text{wmAa}+\text{wmAA}-C\,\text{wmAA})+C\,\text{wfAA wmaa}\,(1+C-2\,C\,\delta)+2\,(-1+C)\,\text{wfAa wmaa}\,(-1+C\,(-1+2\,\delta))\big)\Big/\big(2\,(-2+C)\,\text{wfaa wmaa}\,(-1+C\,\delta)\big)$$

Note that under obligate outcrossing, $\lambda^{(k)}_{q=\hat{q}}$ for the boundary equilibria ($\hat{q}=0, \hat{q}=1$) reduces to the standard sex-averaged relative fitness of heterozygotes.

```
D[qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa] /. C → 0, q] /. q → 0 //
  FullSimplify
D[qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa] /. C → 0, q] /. q → 1 //
  FullSimplify
```

$$\frac{1}{2}\left(\frac{\text{wfAa}}{\text{wfAA}}+\frac{\text{wmAa}}{\text{wmAA}}\right)$$

$$\frac{1}{2}\left(\frac{\text{wfAa}}{\text{wfaa}}+\frac{\text{wmAa}}{\text{wmaa}}\right)$$

```
Clear[λk0, λk1]
```

- Sexually antagonistic fitness selection (see Table 1 in the main text)

```
λkSAq0 :=
  D[qPr[q, C, δ, 1, (1 - hf sf), (1 - sf), (1 - sm), (1 - hm sm), 1], q] /. q → 0 //
    Expand // FullSimplify
λkSAq1 := D[qPr[q, C, δ, 1, (1 - hf sf), (1 - sf), (1 - sm), (1 - hm sm), 1], q] /.
    q → 1 // FullSimplify
λkSAq0
λkSAq1
```

$$\left(2\left(-2+sm+hm\,sm+hf\left(sf-sf\,sm\right)\right)+C\left(2+sf+sm-4\,hm\,sm-sf\,sm+4\left(-1+hf\,sf\right)(-1+sm)\,\delta\right)+C^2\left(\left(-1+2\,hm\right)sm+2\,(-1+sm)\,\delta-\left(-1+2\,hf\right)sf\,(-1+sm)\,(-1+2\,\delta)\right)\right)\Big/\left(2\,(-2+C)\,(-1+sm)\,(-1+C\,\delta)\right)$$

$$\left(2\left(-2+hm\,sm+sf\left(1+hf-hm\,sm\right)\right)+C\left(2+sm-4\,hm\,sm+4\,\delta+sf\left(-3-sm+4\,hm\,sm-4\,hf\,\delta\right)\right)+C^2\left(-sm+2\,hm\,sm-2\,\delta+sf\left(1-2\,hf+sm-2\,hm\,sm+4\,hf\,\delta\right)\right)\right)\Big/\left(2\,(-2+C)\left(-1+sf\right)(-1+C\,\delta)\right)$$

Note that the invasion conditions for sexually antagonistic alleles under partial selfing given in Olito (2017) can be recovered from these results

```
λkSAq0 /. δ → 0 // FullSimplify
Solve[% == 1, sf] // FullSimplify
λkSAq1 /. δ → 0 // FullSimplify
Solve[% == 1, sf] // FullSimplify
```

$$\frac{1}{2\,(-2+C)\,(-1+sm)}\left(4-C\left(2+sf\right)+2\,hf\,sf\,(-1+sm)-2\left(1+hm\right)sm+C\left(-1+4\,hm+sf\right)sm+C^2\left(sm-2\,hm\,sm+sf\left(-1+2\,hf+sm-2\,hf\,sm\right)\right)\right)$$

$$\left\{\left\{sf\to-\frac{(-1+C)\left(2-C+2\,(-1+C)\,hm\right)sm}{(1+C)\left(-C+2\,(-1+C)\,hf\right)(-1+sm)}\right\}\right\}$$

$$\frac{4-2\,C-2\,sf+3\,C\,sf-C^2\,sf-2\,hf\,sf+2\,C^2\,hf\,sf+(-1+C)\left(-C+2\,(-1+C)\,hm\right)\left(-1+sf\right)sm}{2\,(-2+C)\left(-1+sf\right)}$$

$$\left\{\left\{sf\to\frac{(-1+C)\left(-C+2\,(-1+C)\,hm\right)sm}{(1+C)\left(2-C+2\,(-1+C)\,hf\right)+(-1+C)\left(-C+2\,(-1+C)\,hm\right)sm}\right\}\right\}$$

```
Manipulate[Plot[{-((-1 + C) (2 - C + 2 (-1 + C) h) sm)/((1 + C) (-C + 2 (-1 + C) h) (-1 + sm)),
   ((-1 + C) (-C + 2 (-1 + C) h) sm)/((1 + C) (2 - C + 2 (-1 + C) h) + (-1 + C) (-C + 2 (-1 + C) h) sm)}, {sm, 0, 1},
  PlotRange → {0, 1}, AxesLabel → {"sm", "sf"}, AspectRatio → 1, Filling → {1 → {2}},
  PlotStyle → {Black, Black, Directive[Blue, Dashed], Directive[Blue, Dashed]}],
 {C, 0, 1}, {{h, 1 / 2}, 0, 1 / 2}]
```



And under obligate outcrossing and additive fitness effects, the standard invasion conditions for separate-sexed species is also recovered

```
λkSAq0 /. C → 0 /. hf → 1 / 2 /. hm → 1 / 2 // FullSimplify
Solve[% == 1, sf] // FullSimplify
λkSAq1 /. C → 0 /. hf → 1 / 2 /. hm → 1 / 2 // FullSimplify
Solve[% == 1, sf] // FullSimplify
```

$$\frac{4 - sf - 3\,sm + sf\,sm}{4 - 4\,sm}$$

$$\left\{\left\{sf \to \frac{sm}{1 - sm}\right\}\right\}$$

$$\frac{4 - 3\,sf - sm + sf\,sm}{4 - 4\,sf}$$

$$\left\{\left\{sf \to \frac{sm}{1 + sm}\right\}\right\}$$

Simplifying $\lambda^{(k)}_{q=\hat{q}}$ under additive fitness

- Expressing fitness of rare homozygotes in terms of rare heterozygotes

For the boundary equilibrium $\hat{q} = 0$, we parameterize fitness using common *AA* homozygotes as the reference genotype.

$w_{\mathrm{AA}} = 1$, $w_{Aa} = 1 - v^{(k)}/2$, $w_{aa} = 1 - v^{(k)}$

Solving $w_{Aa} = 1 - v^{(k)}/2$ for $v^{(k)}$, and substituting into the expression for w_{aa} yields: $w_{aa} = 1 - 2\,(1 - w_{Aa})$

For the boundary equilibrium $\hat{q} = 1$, we re-parameterize fitness to use common *aa* homozygotes as the reference genotype, but maintain additive fitness and the fitness differences between genotypes:

$w_{AA} = 1 + v^{(k)}$, $w_{Aa} = 1 + v^{(k)}/2$, $w_{aa} = 1$

Again solving w_{Aa} for $v^{(k)}$, and substituting into the expression for w_{aa} yields: $w_{aa} = 2\,(w_{Aa} - 1) + 1$

- Equal selection through each sex

```
λk0 := D[qPr[q, C, δ, wAA, wAa, waa, wAA, wAa, waa], q] /. q → 0 // FullSimplify
λk1 := D[qPr[q, C, δ, wAA, wAa, waa, wAA, wAa, waa], q] /. q → 1 // FullSimplify
```

First, $\lambda^{(k)}_{q=\hat{q}}$ for the boundary equilibrium $\hat{q} = 0$

```
λk0 // Expand
% /. waa → 1 - 2 (1 - wAa) // Simplify // Expand
% /. C → 0
```

$$\frac{C\,waa}{2\,wAA - C\,wAA} + \frac{2\,wAa}{2\,wAA - C\,wAA} - \frac{2\,C\,wAa}{2\,wAA - C\,wAA}$$

$$\frac{C}{(-2 + C)\,wAA} - \frac{2\,wAa}{(-2 + C)\,wAA}$$

$$\frac{wAa}{wAA}$$

Second, $\lambda^{(k)}_{q=\hat{q}}$ for the boundary equilibrium $\hat{q} = 1$

```
λk1 // Expand
% /. wAA → 2 (wAa - 1) + 1 // Simplify // Expand
% /. C → 0
```

$$\frac{2\,\text{wAa}}{2\,\text{waa} - \text{C}\,\text{waa}} - \frac{2\,\text{C}\,\text{wAa}}{2\,\text{waa} - \text{C}\,\text{waa}} + \frac{\text{C}\,\text{wAA}}{2\,\text{waa} - \text{C}\,\text{waa}}$$

$$\frac{\text{C}}{(-2 + \text{C})\,\text{waa}} - \frac{2\,\text{wAa}}{(-2 + \text{C})\,\text{waa}}$$

$$\frac{\text{wAa}}{\text{waa}}$$

```
Clear[λk0, λk1]
```

- Sex-specific selection

```
λk0 := D[qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa] /. δ → 0, q] /. q → 0 //
  FullSimplify
λk1 := D[qPr[q, C, δ, wfAA, wfAa, wfaa, wmAA, wmAa, wmaa] /. δ → 0, q] /. q → 1 //
  FullSimplify
```

First, $\lambda^{(k)}_{q=\hat{q}}$ for the boundary equilibrium $\hat{q} = 0$

```
λk0
λk0 /. wfaa → 1 - 2 (1 - wfAa) /. wmaa → 1 - 2 (1 - wmAa) // FullSimplify
x = Collect[%, {wfAa, wmAa}]
Expand[x[[3]]] == Expand[(1 - C) C/(2 (-2 + C) wmAA) + C (1 + C)/(2 (-2 + C) wfAA)]
(1 - C) C/(2 (-2 + C) wmAA) + C (1 + C)/(2 (-2 + C) wfAA) /. wfAA → 1 /. wmAA → 1 // FullSimplify
λ0simplified = x[[1 ;; 2]] + %
```

$$\Big((-1 + \text{C})\,\text{wfAA}\,(\text{C}\,(\text{wmaa} - 2\,\text{wmAa}) + 2\,\text{wmAa}) - (1 + \text{C})\,\big(\text{C}\,(\text{wfaa} - 2\,\text{wfAa}) + 2\,\text{wfAa}\big)\,\text{wmAA}\Big) \Big/ \big(2\,(-2 + \text{C})\,\text{wfAA}\,\text{wmAA}\big)$$

$$\frac{-(-1 + \text{C})\,\text{wfAA}\,(\text{C} - 2\,\text{wmAa}) + (1 + \text{C})\,(\text{C} - 2\,\text{wfAa})\,\text{wmAA}}{2\,(-2 + \text{C})\,\text{wfAA}\,\text{wmAA}}$$

$$-\frac{(1 + \text{C})\,\text{wfAa}}{(-2 + \text{C})\,\text{wfAA}} - \frac{(1 - \text{C})\,\text{wmAa}}{(-2 + \text{C})\,\text{wmAA}} + \frac{(1 - \text{C})\,\text{C}\,\text{wfAA} + \text{C}\,(1 + \text{C})\,\text{wmAA}}{2\,(-2 + \text{C})\,\text{wfAA}\,\text{wmAA}}$$

True

$$\frac{\text{C}}{-2 + \text{C}}$$

$$\frac{\text{C}}{-2 + \text{C}} - \frac{(1 + \text{C})\,\text{wfAa}}{(-2 + \text{C})\,\text{wfAA}} - \frac{(1 - \text{C})\,\text{wmAa}}{(-2 + \text{C})\,\text{wmAA}}$$

Then, $\lambda^{(k)}_{q=\hat{q}}$ for the boundary equilibrium $\hat{q} = 1$

```
λk1
λk1 /. wfAA → 2 (wfAa - 1) + 1 /. wmAA → 2 (wmAa - 1) + 1 // FullSimplify
Collect[Expand[%], {wfAa, wmAa}]
```

$$-\left(\left(-2\left(-1+C^2\right)\text{wfAa wmaa}+C\,(1+C)\,\text{wfAA wmaa}+2\,\text{wfaa wmAa}+C\,\text{wfaa}\,(2\,(-2+C)\,\text{wmAa}+\text{wmAA}-C\,\text{wmAA})\right)\Big/\left(2\,(-2+C)\,\text{wfaa wmaa}\right)\right)$$

$$\left(C^2\,(-\text{wfaa}+\text{wmaa})-2\,(\text{wfAa wmaa}+\text{wfaa wmAa})+C\,(\text{wfaa}+\text{wmaa}-2\,\text{wfAa wmaa}+2\,\text{wfaa wmAa})\right)\Big/\left(2\,(-2+C)\,\text{wfaa wmaa}\right)$$

$$\frac{C}{2\,(-2+C)\,\text{wfaa}}+\frac{C^2}{2\,(-2+C)\,\text{wfaa}}+\left(-\frac{1}{(-2+C)\,\text{wfaa}}-\frac{C}{(-2+C)\,\text{wfaa}}\right)\text{wfAa}+\frac{C}{2\,(-2+C)\,\text{wmaa}}-\frac{C^2}{2\,(-2+C)\,\text{wmaa}}+\left(-\frac{1}{(-2+C)\,\text{wmaa}}+\frac{C}{(-2+C)\,\text{wmaa}}\right)\text{wmAa}$$

Which simplifies to:

$$\frac{1+C}{2-C}\frac{w^f_{\mathrm{Aa}}}{w^f_{\mathrm{aa}}}+\frac{1-C}{2-C}\frac{w^m_{\mathrm{Aa}}}{w^m_{\mathrm{aa}}}-\frac{C}{2-C}$$

```
Clear[λk0, λk1]
```

References

A. Olito C. 2017 Consequences of genetic linkage for the maintenance of sexually antagonistic polymorphism in hermaphrodites. *Evolution* **71**,458-464.

B. Jordan CY, Connallon T. 2014 Sexually antagonistic polymorphism in simultaneous hermaphrodites. *Evolution* **68**, 3555-3569.

C. Charlesworth B, Charlesworth D. 2010 *Elements of Evolutionary Genetics*. Pp. 77 - 80; 589 - 590. Roberts and Company, Greenwood Village, CO.